(% i5) /* Declare and assume*/
declare(g, constant)$
declare(Ω, constant)$
declare(ϕ, constant)$
declare(h, constant)$
assume(ϕ >= - %pi / 2, ϕ <= %pi / 2, g > 0, Ω > 0, t >= 0, h > 0)$

(% i6) /* First equation*/
eq1: 2 * Ω * sin(ϕ) * diff(y(t), t, 1) - 2 * Ω * cos(ϕ) * diff(z(t), t, 1) + Ω ^ 2 * x(t) = diff(x(t), t, 2);

$$\text{(eq1)}\quad -\left(2\Omega\cos\left(\phi\right)\left(\frac{d}{dt}\,z(t)\right)\right)+2\Omega\sin\left(\phi\right)\left(\frac{d}{dt}\,y(t)\right)+\Omega^2\,x(t)=\frac{d^2}{dt^2}\,x(t)$$

(% i7) /* Second equation*/
eq2: - 2 * Ω * sin(ϕ) * diff(x(t), t, 1) + Ω ^ 2 * (sin(ϕ)) ^ 2 * y(t) - Ω ^ 2 * sin(ϕ) * cos(ϕ) * z(t) = diff(y(t), t, 2);

(eq2)

$$-\left(2\Omega\sin\left(\phi\right)\left(\frac{d}{dt}\,x(t)\right)\right)-\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)z(t)+\Omega^2\sin\left(\phi\right)^2y(t)=\frac{d^2}{dt^2}\,y(t)$$

(% i8) /* Third equation*/
eq3: - g + 2 * Ω * cos(ϕ) * diff(x(t), t, 1) + Ω ^ 2 * (cos(ϕ)) ^ 2 * z(t) - Ω ^ 2 * sin(ϕ) * cos(ϕ) * y(t) = diff(z(t), t, 2);

(eq3)

$$2\Omega\cos\left(\phi\right)\left(\frac{d}{dt}\,x(t)\right)+\Omega^2\cos\left(\phi\right)^2z(t)-\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)y(t)-g=\frac{d^2}{dt^2}\,z(t)$$

(% i14) /* Values at t = 0*/
atvalue(x(t), t=0, 0)$
atvalue(y(t), t=0, 0)$
atvalue(z(t), t=0, h)$
atvalue(diff(x(t), t, 1), t=0, 0)$
atvalue(diff(y(t), t, 1), t=0, 0)$
atvalue(diff(z(t), t, 1), t=0, 0)$

(% i16) /* Declare s complex*/
declare(s, complex)$
assume(realpart(s) > 0)$

(% i19) /* Apply Laplace transform to equations*/
eq1L: laplace(eq1, t, s);
eq2L: laplace(eq2, t, s);
eq3L: laplace(eq3, t, s);

(eq1L)

$$-\left(2\Omega\cos\left(\phi\right)\left(s\,laplace\left(z(t)\,,t\,,s\right)-h\right)\right)+2\Omega\sin\left(\phi\right)s\,laplace\left(y(t)\,,t\,,s\right)+\Omega^2\,laplace\left(x(t)\,,t\,,s\right)=s^2\,laplace\left(x(t\right.$$

(eq2L)
$$-\left(\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)laplace\left(z(t)\,,t\,,s\right)\right)+\Omega^2\sin\left(\phi\right)^2 laplace\left(y(t)\,,t\,,s\right)-2\Omega\sin\left(\phi\right)s\,laplace\left(x(t)\,,t\,,s\right)=s^2\,lapl$$

(eq3L)
$$\Omega^2\cos\left(\phi\right)^2 laplace\left(z(t)\,,t\,,s\right)-\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)laplace\left(y(t)\,,t\,,s\right)+2\Omega\cos\left(\phi\right)s\,laplace\left(x(t)\,,t\,,s\right)-\frac{g}{s}=s^2\,lapl$$

(% i22)
```
/* Simplify notation*/
eq1L: subst([
laplace(x(t), t, s) = X(s),
laplace(y(t), t, s) = Y(s),
laplace(z(t), t, s) = Z(s)
], eq1L);
eq2L: subst([
laplace(x(t), t, s) = X(s),
laplace(y(t), t, s) = Y(s),
laplace(z(t), t, s) = Z(s)
], eq2L);
eq3L: subst([
laplace(x(t), t, s) = X(s),
laplace(y(t), t, s) = Y(s),
laplace(z(t), t, s) = Z(s)
], eq3L);
```

(eq1L) $-\left(2\Omega\cos\left(\phi\right)\left(s\,Z(s)-h\right)\right)+2\Omega\sin\left(\phi\right)s\,Y(s)+\Omega^2\,X(s)=s^2\,X(s)$

(eq2L)
$$-\left(\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)Z(s)\right)+\Omega^2\sin\left(\phi\right)^2 Y(s)-2\Omega\sin\left(\phi\right)s\,X(s)=s^2\,Y(s)$$

(eq3L)
$$\Omega^2\cos\left(\phi\right)^2 Z(s)-\Omega^2\cos\left(\phi\right)\sin\left(\phi\right)Y(s)+2\Omega\cos\left(\phi\right)s\,X(s)-\frac{g}{s}=s^2\,Z(s)-hs$$

(% i23)
```
/* Find X(s), Y(s) and Z(s)*/
solL: factor(trigsimp(solve([eq1L, eq2L, eq3L], [X(s), Y(s), Z(s)])[1]));
```

(solL)

$$[\,X(s)=-\left(\frac{2\Omega\,\left(h\Omega^2-g\right)\cos(\phi)}{\left(s^2+\Omega^2\right)^2}\right),Y(s)=-\left(\frac{\Omega^2\cos(\phi)\sin(\phi)\left(hs^4-h\Omega^2s^2+3gs^2+g\Omega^2\right)}{s^3\left(s^2+\Omega^2\right)^2}\right),Z(s)=(\,hs^6+$$

(% i24) /* Find x(t) by inverse Laplace transform*/
xt: factor(trigsimp(ilt(rhs(solL[1]), s, t)));

(xt) $$-\left(\frac{\left(h\Omega^2-g\right)\cos(\phi)\left(\sin(\Omega t)-\Omega t\cos(\Omega t)\right)}{\Omega^2}\right)$$

(% i25) /* Find y(t) by inverse Laplace transform*/
yt: factor(trigsimp(ilt(rhs(solL[2]), s, t)));

(yt)
$$-\left(\left(\cos(\phi)\sin(\phi)\left(2h\Omega^3t\sin(\Omega t)-2g\Omega t\sin(\Omega t)+2h\Omega^2\cos(\Omega t)-2g\cos(\Omega t)+g\Omega^2t^2-2h\Omega^2+2g\right)\right)/\left(2\Omega^2\right)\right)$$

(% i26) /* Find z(t) by inverse Laplace transform*/
zt: factor(trigsimp(ilt(rhs(solL[3]), s, t)));

(zt)
$$(\,2h\Omega^3\cos(\phi)^2t\sin(\Omega t)-2g\Omega\cos(\phi)^2t\sin(\Omega t)+2h\Omega^2\cos(\phi)^2\cos(\Omega t)-2g\cos(\phi)^2\cos(\Omega t)+g\Omega^2\cos(\phi)^2t^2-g$$

(% i27) taylor(xt, Ω, 0, 3);

(% o27/T/) $$\frac{g\cos(\phi)t^3\Omega}{3}-\frac{\left(g\cos(\phi)t^5+10h\cos(\phi)t^3\right)\Omega^3}{30}+\ldots$$

(% i28) taylor(yt, Ω, 0, 4);

(% o28/T/)
$$-\left(\frac{\left(g\cos(\phi)\sin(\phi)t^4+4h\cos(\phi)\sin(\phi)t^2\right)\Omega^2}{8}\right)+\frac{\left(g\cos(\phi)\sin(\phi)t^6+18h\cos(\phi)\sin(\phi)t^4\right)\Omega^4}{144}+\ldots$$

(% i29) taylor(zt, Ω, 0, 2);

(% o29/T/) $$-\left(\frac{gt^2-2h}{2}\right)+\frac{\left(g\cos(\phi)^2t^4+4h\cos(\phi)^2t^2\right)\Omega^2}{8}+\ldots$$

(% i30) D_east: factor(subst(t = sqrt(2 * h / g), xt));

(D_ east) $$-\left(\frac{\left(h\Omega^2-g\right)\left(\sqrt{g}\sin\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right)-\sqrt{2}\sqrt{h}\Omega\cos\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right)\right)\cos(\phi)}{\sqrt{g}\Omega^2}\right)$$

(% i31) taylor(D_east, Ω, 0, 3);

(% o31/T/) $$\frac{\sqrt{2}^{3}\sqrt{h}^{3}\cos(\phi)\Omega}{3\sqrt{g}} - \frac{\sqrt{2}^{5}\sqrt{h}^{5}\cos(\phi)\Omega^{3}}{5\sqrt{g}^{3}} + \ldots$$

(% i32) D_north: factor(subst(t = sqrt(2 * h / g), yt));

(D_ north)

$$-(((\sqrt{2}h^{\frac{3}{2}}\Omega^{3}\sin\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right) - \sqrt{2}g\sqrt{h}\Omega\sin\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right) + \sqrt{g}h\Omega^{2}\cos\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right) - g^{\frac{3}{2}}\cos\left(\frac{\sqrt{2}\sqrt{h}\Omega}{\sqrt{g}}\right) + g^{\frac{3}{2}}$$

(% i33) taylor(D_north, Ω, 0, 4);

(% o33/T/) $$-\left(\frac{3h^{2}\cos(\phi)\sin(\phi)\Omega^{2}}{2g}\right) + \frac{5h^{3}\cos(\phi)\sin(\phi)\Omega^{4}}{9g^{2}} + \ldots$$